Iterative Method of Evaluating Local Electron Density in Asymmetrical Plasma Distributions

N.Gottardi+

IPP III/39 December 1977

MAX-PLANCK-INSTITUT FÜR PLASMAPHYSIK

8046 GARCHING BEI MÜNCHEN

MAX-PLANCK-INSTITUT FÜR PLASMAPHYSIK

GARCHING BEI MÜNCHEN

Iterative Method of Evaluating Local Electron Density in Asymmetrical Plasma Distributions

N.Gottardi[+]

IPP III/39 December 1977

[+] Gastforscher vom Istituto di Macchine-Politecnico di Milano

Die nachstehende Arbeit wurde im Rahmen des Vertrages zwischen dem Max-Planck-Institut für Plasmaphysik und der Europäischen Atomgemeinschaft über die Zusammenarbeit auf dem Gebiete der Plasmaphysik durchgeführt.

IPP III/39 N. Gottardi

Iterative Method of Evaluating Local Electron Density in Asymmetrical Plasma Distributions

Abstract

Obtaining electron density profiles from integrated measurements (e.g. those made with a multi-channel interferometer), in the case of asymmetrical plasma distributions often entails serious difficulties. This report describes a method of obtaining good results when the profiles have only one maximum, possibly shifted along the equatorial plane. It is then extended to more complicated density profiles.

Table of Contents

I. Introduction

In the following, we shall consider the problems of deriving the electron density profile of a plasma column from a set of integrated measurements (e.g. those obtained with a multi-channel interferometer).

We shall assume that the set of experimental data is interpolated, e.g. by a spline fit, to give a curve; in the case of microwave interferometry, as in Fig.2, the curve would be the phase shift of the waves $\Delta\varphi$ as a function of x.

In order to determine the mathematical problem, some assumptions about the form of the isodensity level lines (hereafter simply called "contour lines") or equivalent assumptions are needed.

The resulting integral equation for the unknown density profile is generally analytically unsolvable; a solution is in fact only possible when the contour lines are concentric circles. After general information about the problem and the presentation of the required hypotheses (Sects. II and III), a simple computer iteration method is described (Sect. IV) and then tested in various situations (Sect. V, VI and VII). The report finishes with a short criticism of the method (Sect. VIII).

II. General remarks on electron density measurements by interferometry

The use of interferometry to measure the electron density is a well-known plasma diagnostics method /1/, /2/. In Fig.1 only an application scheme is shown. Here it should be remembered that:

a) an interferometer gives the total phase shift of a wave travelling through a plasma, with respect to the phase



Fig.1: Application scheme of microwave interferometer.



Fig.2: Phase shift and relative mean density of discharge N.10.585 of the Pulsator tokamak 50 ms after the neutral gas puffing.

of a wave that travels along an unperturbed path;

b) except for special cases, only information on the mean density is obtained;

c) in order to obtain some more detail, multi-channel interferometers are used.

In Fig.2, a typical multi-channel interferometer output /3/ is represented. This is the total phase shift and the mean density profile in a Pulsator tokamak discharge.

III. Basic assumptions

If we use an electromagnetic wave of wavelength λ to scan a plasma in the hypothesis that the dielectric properties of the plasma do not appreciably change over a length λ, the total phase shift $\Delta\varphi_S$ that the wave undergoes on passing through the plasma along a path S is given in the adiabatic approximation /1/, /4/ by

$$\Delta\varphi_S = (2\pi/\lambda)\cdot\int_S \left\{1-\left[1-\frac{n(S)}{n_c}\right]^{\frac{1}{2}}\right\} ds \quad , \qquad \text{(III.1)}$$

where n(s) is the local electron density and n_c is the critical density for the wave:

$$n_c = \varepsilon_o m \omega^2/e^2 \, .$$

Here ε_o is the permittivity of the free space; n and e are the mass and the charge of the electron respectively, and $\omega = 2\pi c/\lambda$ is the angular frequency of the wave.

Equation (III.1) can be considered as an integral equation for the unknown density profile. One usually takes for the measurements wave whose frequency, while allowing good sensitivity at low densities, is such that the associated critical density n_c is sufficiently greater than the plasma density n(s). Under this assumption, we can expand the root in Eqn.(III.1), thus obtaining

$$\Delta\varphi_s \cong \frac{\omega}{2cn_c} \cdot \int_s n(s)\,ds \ . \qquad \text{(III.2)}$$



Fig.3: Three-dimentional representation and contour lines of a plasma with cylindrical symmetry.

If the plasma has cylindrical symmetry as in Fig.3, Eqn.(III.2) can be rewritten as

$$\Delta\varphi(x) \cong \frac{\omega}{c n_c} \cdot \int_x^a \frac{n(r)\, r}{\sqrt{r^2 - x^2}}\, dr \ . \qquad \text{(III.3)}$$

This is an Abel integral equation; it can be solved /5/ to yield

$$n(r) = -\frac{2 c n_c}{\pi \omega} \cdot \int_r^a \frac{\Delta\varphi'(x)}{\sqrt{x^2 - r^2}}\, dx \ .$$

This is, of course, a rather special situation, in most cases the plasma column is asymmetrical and then the only information that we can obtain wihtout further hypotheses is the mean value of the density. In fact, if we know the total path S of the wave inside the plasma, the mean density $\bar{n}_S$ is

$$\bar{n}_S = \left(\int_S n(s)\, ds \right) / S \ ,$$

and from Eqn.(III.2) we obtain

$$\bar{n}_S = (2 c n_c / \omega)\, \Delta\varphi_S / S \ . \qquad \text{(III.4)}$$

However, as the most interesting quantity is the density profile, in the event of asymmetrical plasma configurations one makes assumptions about the most probable spatial displacement, either introducing appropriate modifications to Eqn.(III.2) /5/ or suitably handling the mean density values obtained from Eqn.(III.4) /6/.



Fig.4: Three-dimensional representation and contour lines of a shifted plasma.

A frequently used hypothesis, justified by the shape of the confining magentic field, is that the lines of equal electron density are circles, not necessarily with the same centre (see Fig.4). With a simple computer program the numerical iteration method described below allows one to get information sufficiently accurate in comparison with the intrinsic total error induced by the basic hypotheses.

The following assumptions are made:

a) Adiabatic approximation;

b) Plasma density much lower than the critical density: $n \ll n_c$;

c) The isodensity curves are symmetric with respect to the X-axis, as shown in Fig.4.

The program, developed in order to obtain the local electron density profile directly from the data yielded by a multi-channel interferometer for a fast plasma diagnostics acquisition system, may also apply to other diagnostics and also in fields other than plasma physics (e.g. analysis of the light emitted from a flame in combustion research).

IV. Establishment of the method

The upper part of Fig.5 shows a three-dimentional model of the electron density distribution with displacement along the X-axis but still with symmetry about the equatorial torus planes. Here the shape of the contour lines has circular symmetry but this, as explained later in Sect.VI is not essential in principle to the application of the method, but it does simplify understanding of the method.

It should also be stated that the phase shift function $\Delta\varphi(x)$ that interpolates the experimental data is, in reality, only a function tabulated with a certain number of points. This means that in the following we refer to $\Delta\varphi(x)$ as a vector whose elements are $\Delta\varphi(x_k)$, where $1 \leq k \leq t$.

Let us now come back to Fig.5 and consider a plasma density cross-section perpendicular to both the equatorial and interferometer planes and passing through an interferometric channel line with abscissa x_k. Let us apply Eqn.(III.2) to the path (-a, +a) in the y direction:



Fig.5: Cross section of a shifted density distribution (top) corresponding to an interferometric channel path (bottom).

$$\Delta\varphi(x_k) = (\omega/2cn_c)\cdot\int_{-a}^{a} n(y)\,dy \quad . \tag{IV.1}$$

The cross-sectional area, but for a constant, represents the total phase shift of the wave for that path.

Approximating this with a figure in steps as shown in Fig.6 and remembering that for the above hypotheses, the shape of the density function n(y), although unknown, is symmetric, we can rewrite Eqn.(IV.1) as

$$\Delta\varphi(x_k)/h \simeq \Delta\Phi_{(1,m)}(x_k) \equiv 2\sum_{i=1}^{m} n_i \Delta l_i \quad , \tag{IV.2}$$

where $\Delta\Phi$ is the area of the approximating figure in steps and $h \equiv \omega/2cn_c$. We can also write (see Fig.6)

$$\Delta\Phi_{(1,m)}(x_k) = \sum_{i=1}^{m} l_i \Delta n_i \tag{IV.3}$$

and, in addition,

$$n_i = \sum_{j=1}^{i} \Delta n_j \quad . \tag{IV.4}$$

In Eqn.(IV.3), the left-hand side is experimentally known; about the quantities on the right-hand side we observe that, since for definition the Δn_i are arbitrary, they are chosen very small to yield a good approximation, and all of the same height Δn except, possibly, the last upper Δn_m. Then Eqn.(IV.4), written at the point of maximum density, becomes



Fig.6: Step approximation of the cross-sectional area of Fig.5



Fig.7: Approximation of the cross sectional area of Fig.5 with steps of equal height Δn.

$$n_m = \sum_{i=1}^{m} \Delta n_i = (m-1)\Delta n + \Delta n_m \ , \qquad \text{(IV.5)}$$

where $\Delta n_m \leq \Delta n$.

We define Δn as a very small fraction of the mean density:

$$\Delta n \equiv \varepsilon \left[\bar{n}(x_k)\right] \ , \qquad \text{(IV.6)}$$

where $\bar{n}$ is derived from Eqn.(III.4), and $\bar{n} \leq n_m$. For the following it will be useful to think of the area of the approximating figure in steps (see Fig.6) as if it were devided in three parts (see. Fig.7);

$$\Delta \Phi_{(1,m)}(x_k) = \Delta \Phi_{(1,q)}(x_k) + \Delta \Phi_{(q+1,m-1)}(x_k) + \ell_m \cdot \Delta n_m \ , \qquad \text{(IV.7)}$$

where

$$\Delta \Phi_{(1,q)}(x_k) \equiv \sum_{i=1}^{q} \ell_i \Delta n \qquad \text{and} \qquad \Delta \Phi_{(q+1,m-1)}(x_k) \equiv \sum_{i=q+1}^{m-1} \ell_i \Delta n \ ;$$

particularly for q = 1 Eqn.(IV.7) becomes

$$\Delta \Phi_{(1,m)}(x_k) = \Delta \Phi_{(1,1)}(x_k) + \Delta \Phi_{(2,m-1)}(x_k) + \ell_m \Delta n_m \qquad \text{(IV.8)}$$

From Fig.8, which represents the density distribution in Fig.5, approximated by a solid with step-like structure, we see that ℓ_1 is the total path of the wave into the plasma:

$$\ell_1 = 2a \qquad \text{(IV.9)}$$



Fig.8: Schematic representation of the electron density distribution of Fig.5 approximated by a solid with step-like structure.

The elements of the cross-section corresponding to abscissa x_k have the same significance as in Figs.6 and 7.

The value of the first term on the right-hand side of Eqn.(IV.8) is easily determined from Eqns.(IV.9) and (IV.6):

$$\Delta\Phi_{(1,1)}(x_k) = \ell_1 \Delta n \quad . \tag{IV.10}$$

We thus write

$$\Delta\Phi_{(1,m)}(x_k) - \Delta\Phi_{(1,1)}(x_k) = \Delta\Phi_{(2,m-1)}(x_k) + \ell_m \Delta n_m \tag{IV.11}$$

or

$$\Delta\Phi_{(2,m)}(x_k) = \Delta\Phi_{(2,2)}(x_k) + \Delta\Phi_{(3,m-1)}(x_k) + \ell_m \Delta n_m \quad . \tag{IV.12}$$

From Eqn.(IV.10) and from the above considerations we know the value of the left-hand side of Eqn.(IV.12), which is of the same form as Eqn.(IV.8). Because of

$$\Delta\Phi_{(2,2)} = \ell_2 \Delta n \ ,$$

if we know the value of ℓ_2 it is possible, with a process similar to that leading from Eqn.(IV.8) to Eqn.(IV.12), to deduce the value of $\Delta\Phi_{(3,m)}(x_k)$.

By induction we rewrite Eqn.(IV.11) in the form

$$\Delta\Phi_{(p,m)}(x_k) - \Delta\Phi_{(p,p)}(x_k) = \Delta\Phi_{(p+1,m-1)}(x_k) + \ell_m \Delta n_m \ , \tag{IV.13}$$

where

$$\Delta\Phi_{(p,p)}(x_k) = \ell_p \Delta n \quad \text{and} \quad 1 \leq p \leq m \quad .$$

If we suppose that we have a method of finding the value of all the other chords $l_3 \ldots l_m$, Eqn.(IV.13) is an iterative expression whose left-hand side, which represents the area of Fig.7 from which one substracts every time the area of the lowest step, is known at every step.

The iteration is terminated when the whole step-shaped area is exhausted. At this point the left-hand side of Eqn.(IV.13) assumes a negative or a zero value since $\Delta n_m \leq \Delta n$.
By means of Eqn.(IV.5), truncated at the $(m-1)^{th}$ term, we now obtain the value of $n_m(x_k)$ with an error of about Δn_m. The problem of deriving the electron density in the equatorial plane of the torus is now reduced to that of finding the chords l_i. As already stated, this problem is not solvable without some assumption about the form of the contour lines.

We shall, however, show in the following that two conditions must always be satisfied, at every plane of our step model, for every form of the contour lines. Let us show this first in the very simple case of a two-step model, as in Fig.9 (the cylindrical form of the disks is not essential, as will be seen).

The lower part of Fig. 10, shows an equatorial section of Fig.9, while the upper part shows the corresponding total phase shift profile. The tabulated total phase shift profile can be written as

$$\Delta\varphi(x_k) = h\left[\ell_1(x_k)n_1 + \ell_2(x_k)\cdot(n_2-n_1)\right] \equiv \Delta\varphi_1(x_k) + \Delta\varphi_2(x_k) \ , \qquad \text{(IV.14)}$$

where $\ell_1(x_k)$ and $\ell_2(x_k)$ are the chords through the points x_k.
It should now be noted that the phase shift due to the lower disk is

$$\Delta\varphi_1(x_k) = h\,\ell_1(x_k)n_1 \ ;$$



Fig.9: Electron density distribuiton model with two cylindrical steps.



Fig.10: Cross section (bottom) of the cylindrical two-step model of Fig.9 and corresponding phase shift (top).

if we subtract this value from $\Delta\varphi(x_k)$ given by Eqn.(IV.14), we get the contribution to $\Delta\varphi(x_k)$ by the upper disk:

$$\Delta\varphi_2(x_k) = h\,\rho_2(x_k)\cdot(n_2 - n_1) \quad .$$

This contribution is, of course, different from zero only when $x_C < x_k < x_D$; a scanning of the points of the tabulated phase shift function, after subtraction, will therefore yield x_C and x_D. It is now clear that the density profile can be derived from the measured values of $\Delta\varphi(x_k)$ only if a relation between (x_C, x_D) and $l_2(x_k)$ is given, possibly on the basis of physical arguments; for example, the isodensity lines are completely defined by (x_C, x_D) if they are circles. Naturally, this rough process may be extended by induction to models with several steps approximating the true electron density distributions. In the general case we also scan the tabulated values as already described to eliminate all those which are zero or negative and to fix the new extremes. The latter give the necessary parameters for determining the chords of the next step. By means of Eqn.(IV.13), where p is the common iteration index for all the x_k, one goes on till the whole $\Delta\Phi_{(1,m)}(x_k)$ is exhausted. The local electron density $n(x_k)$ is obtained from Eqn.(IV.5), whereby it has to be remembered that m assumes the value of the common iteration index p just when the $\Delta\Phi_{(p,m)}(x_k)$ become zero or negative.

Appendix I presents the basic program in Fortran which performs the foregoing for the next examples.

V. Testing with assigned density distributions with only one maximum

It is assumed that the electron density distribution is given; we then evaluate the phase shift and finally deduce from the

phase shift the density profile by the method just described. Let us assume that the electron density distribution is

$$n(x,y) = -(\alpha/\rho^2)\cdot(x^2+y^2-\rho^2) \quad ; \tag{V.1}$$

hence the profile is

$$n(x) = -(\alpha/\rho^2)\cdot(x^2-\rho^2) \quad , \tag{V.2}$$

where $\rho \geq (x^2+y^2)^{\frac{1}{2}}$ is the plasma radius and α is the maximum of $n(x)$.

The corresponding phase shift is given by Eqn.(III.2). The function $\Delta\varphi(x)$ is then tabulated with 1000 points, and Δn (see Eqn.IV.6) is chosen as 1/500 of the maximum of the mean density profile obtained by means of Eqn.(III.4).

We now apply the method described in Sect. IV to the phase shift given by Eqn.(III.2). The result is given in Fig.11. There are no appreciable differences between the assigned profile and the result. The error, which may be shown by a print-out, is $< 0.1\,\%$.

As another example, we now consider the highly asymmetric distribution

$$\rho^2 n(x,y) = -\alpha\left\{\left[x+\delta n(x,y)\right]^2+y^2-\rho^2\right\} \quad , \tag{V.3}$$



Fig.11: Assigned density profile as given by Eqn.(V.2) with: $\alpha = 1.5 \cdot 10^{14}$ cm^{-3}, $\rho = 11$ cm. The inverted profile practically coincides with the assigned profile.



Fig.12: Assigned density profile as given by Eqn.(V.4) with: $\alpha = 1.5 \cdot 10^{14}$ cm^{-3}, $\rho = 11$ cm and $\delta = 0.3$ cm^4. As in the symmetrical case, the inverted profile practically coincides with the assigned profile.



Fig.13: The assigned density profile is step-like. The inverted profile exhibits a little erosion only at the step corners.

where δ is a shift parameter; the corresponding profile in the equatorial plane is

$$\rho^2 n(x) = -\alpha \left\{ \left[x + \delta n(x) \right]^2 - \rho^2 \right\} . \qquad (V.4)$$

In Fig.12 one sees the assigned profile compared with results obtained again with 1000 points. The error is $< 0.1\%$.

Finally, to test the method in a very abnormal situation, we start with a step-like distribution. The given density profile, the phase shift and the evaluated profile are drawn in Fig.13: only the step corners have undergone a little erosion.

VI. Testing with assigned density distribution with several maxima

The experimental phase shift is often more complex than in the previous examples.

A frequently observed phase shift is of the type illustrated in Fig.14 A. The usual hypothesis is that it is due to a plasma distribution of the type shown in Fig.14B; in fact, it could also derive from a distribution of the type given in Fig.14C.

VI.a Two-hill model

As already stated, circular symmetry of the contour lines is not necessary for the method described in Sect.IV; in fact, all the expressions in Sect.IV are written for a cross-section through a general point X_k; what is needed is the shape of the contour lines.



Fig.14: The density profile sketched in (A) may give rise to ambiguous interpretation: in fact it may belong to both distributions sketched in (B) and (C).
The representation (D) is the approximation assumed in the case of hill-shaped distribution of (B).



Fig.15: Application scheme of the proposed method for a two-hill shaped distribution.

It is assumed that the two-hill model in Fig.14B can be approximated with a step-like solid such as sketched in Fig.14D we then apply the same method as for the one-maximum distribution, but with the difference that after subtraction of the contribution of a step we scan all the points (not only the



Fig.16: Except for the shape this is a case analogous to that of Fig.13. Here also the inverted profile exhibits a little erosion at the step corners.

lateral ones) to find a possible "valley" (see Fig.15A); if we find one, we proceed to exhaust the first area shown in Fig.15B, and then the other one too (see Fig.15C). An extension to distributions with more than two maxima is easy; however, it must then be remembered, and anyhow cases where the valleys are too deep, that the intrinsic error of the assumed step model becomes much larger than that introduced by the iteration method. Since a two-hill model is very complex, we consider the model shown in Fig.14D; the results are presented in Fig.16

VI.b Crater model

Let us imagine a very simple "crater" distribution: a disk with a hole whose cross-section is shown in the lower part of Fig.17. We rewrite Eqn.(IV.14) for this case:

$$\Delta\varphi(x_k) = \Delta\varphi_1(x_k) - \Delta\varphi_2(x_k)\,. \qquad \text{(VI.1)}$$

To find the points X_C and X_D, we use the same process described after Eqn.(IV.14), noting that, for the negative sign in Eqn.(VI.1), we must terminate scanning at the first two negative points; after this, we restore the original cross-section by subtracting from the right place in the whole disk the step corresponding to the hole. The general case is given in Fig.18: the profile is restored till all the points of the tabulated phase shift function become negative (see Fig.18A); one then subtracts from it the profile due to the negative area (Fig.18B) treated as in the model described in Sect.V.

As an example we start from the following distribution model:

$$\rho^2\left(1+\gamma\left\{\left[x+\delta n(x,y)\right]^2+y^2\right\}\right)n(x,y) = -\alpha\gamma\left\{\left[x+\delta n(x,y)\right]^2+y^2\right\}^2 + \qquad \text{(VI.2)}$$
$$+\left(\alpha\gamma\rho^2+\beta-\alpha\right)\left\{\left[x+\delta n(x,y)\right]^2+y^2\right\}+\rho^2(\alpha-\beta)\,,$$



Fig.17: Cross section of a single step "crater" (bottom) and corresponding phase shift profile (top)



Fig.18: Application scheme of the proposed method for a crater-shaped distribution.



Fig.19: Assigned density profile as given by Eqn.(IV.3) with: $\alpha = 2.5 \cdot 10^{14}$ cm^{-3}, $\beta = 2 \cdot 10^{14}$cm^{-3}, $\gamma = .15$cm^{-2}, $\delta = -.1$cm^4 and $\rho = 11$ cm.
As in the paraboloid distribution case, the inverted density profile practically coincides with the assigned profile.

and profile

$$\rho^2\left\{1+\gamma\left[x+\delta n(x)\right]^2\right\}n(x) = -\alpha\gamma\left[x+\delta n(x)\right]^4 + \\ +(\alpha\gamma\rho^2+\beta-\alpha)\left[x+\delta n(x)\right]^2+\rho^2(\alpha-\beta), \qquad \text{(VI.3)}$$

where α, β and γ are parameters related to the shape of the crater, δ is a shift parameter, and ρ is the radius of the plasma. For this case a more general version of the program was used; it is given in Appendix II and can also be applied to the preceding cases. the results are presented in Fig.19. The error, although not visible in Fig.19, is a little greater than in the paraboliod distribution case owing to the longer iteration. For the sake of brevity the case of a symmetrical crater has not been reported; this gives better results than the asymmetrical case.

VII. Practical application

The time evolution of a plasma profile in a real case is represented in Fig.20.; shot no. 12.792 in the Pulsator tokamak. Each profile was evaluated assuming the hill model described in Sect.IV.a, starting from the corresponding phase shift of the type shown in Fig.21 for the time 40 ms. We note that the phase shift is known at nine points: at the positions of the seven interferometric channels and at the two extreme zero points. The interpolating phase shift function is obtained by means of a cubic spline fit with arbitrary end conditions /7/.

Owing to the small number of experimental values, the interpolation introduces errors which are comparable with the errors due to the choice of the form of the contour lines. In the case of Fig.22, the proposed method was applied with two different heights of the steps (see Eqn.(IV.6). It is found that in the



Fig.20: Time evolution of the plasma profile of discharge No.12.792 of the Pulsator tokamak. The time is counted after the neutral gas puffing.



Fig.21: Detail of the process for obtaining the plasma profiles of Fig.20. Unlike in the text, there are only 7 experimental points: the information relating to the channels 1 and 7 was lost owing to diffraction problems.



Fig.22: Inverted density profiles obtained with two different amplitudes of Δn: (a) with 1/500 of the maximum of the mean density and (b) with only 1/10.

case of a rough approximation we also get good reliability in the mean shape of the density profile. This allows us, when good detail is not required, to save both time and computer space.

VIII. Concluding remarks

As shown above, the proposed method gives very good results in the case of symmetrical and asymmetrical paraboloid distributions; it also gives good results in the symmetrical and asymmetrical crater distributions. In the case of distributions with more than one maximum the results are satisfactory.

It should be noted that this method can be used in an automatic data acquisition system.

References

/1/ M.A.Heald, C.B.Wharton:
Plasma Diagnostics with Microwaves.
John Wiley and Sons, Inc., New York, 1965.

/2/ G.Lisitano:
Microwave-Interferometry
Scuola di Perfezionamento in Fisica Atomica e Nucelare
(1966:67). Universita degli Studi, Milano, 1966.

/3/ G.Lisitano, E.Rossetti, S.Corti:
Contribution to Max-Planck-Institut für Plasmaphysik,
Jahresbericht 1972, IPP München 1972.

/4/ H.Hermansdorfer:
Microwave Diagnostic Techniques
Contribution to Plasma Diagnostics
Edited by W.Lochte-Holtgreven; North-Holland publishing
company, Amsterdam, 1968.

/5/ E.P.Gorbunov, G.N.Ploskirev:
5th Symposium of Fusion Technology; St.Caterhine's College,
Oxford, 1968.

/6/ G.Lisitano, B.Cannici, N.Gottardi, P.Morandi:
Course on Plasma Diagnostics and Data Acquisition Systems.
Edited by A.Eubank and E.Sindoni. Editrice compositori,
Bologna, 1975, p.17

/7/ Internation Mathematical and Statistical Library.
IMSL Inc., Houston, 1977.

APPENDIX I

```
      SUBROUTINE INVERS(X,Y,N,AK,YNVERS)
C************** INVERSION OF HILL-SHAPED DENSITY PROFILES **************
      DIMENSION X(1000),Y(1000),YNVERS(1000)
C     **(X,Y): TABULATION OF THE PHASE SHIFT.
C     **N: NUMBER OF THE TABULATION POINTS.
C     **AK: NORMALISATION CONSTANT.
C     **YNVERS: INVERTED DENSITY.
      Y(1)=0.
      Y(N)=0.
      YMAX=0.
      XMAX=0.
C     **YMAX: MAXIMUM OF THE PHASE SHIFT.
C     **XMAX: ABSCISSA OF THE MAXIMUM.
      DO 1 I=1,N
      YNVERS(I)=0.
      IF(YMAX.LT.Y(I)) GO TO 2
      GO TO 1
 2    YMAX=Y(I)
      XMAX=X(I)
 1    CONTINUE
C     **DETERMINATION OF THE STEP THICKNESS "AN".
C***************************
      NSTEP=500
C***************************
C     **NSTEP: "MINIMUM" NUMBER OF STEPS.
C     **AN: THICKNESS OF LOCAL DENSITY DISK.
      PROD=X(1)*X(N)
      SUMM=X(1)+X(N)
      AN=YMAX/(NSTEP*AK*SQRT(-XMAX**2-PROD+XMAX*SUMM))
      AN2=YMAX/10000.
C*********************************************************************
C     **BEGINNING OF THE "SLICING".
      IBEG=1
      IEND=N
C     **IBEG,IEND: FIRST AND LAST POINT OF A SINGLE SLICE.
C***************************
      KBREIT=10
C***************************
C     ** KBREIT: WIDTH OF THE THE SMALLEST AREA TO BE CONSIDERED; I.E.IF
C     ** AFTER A NUMBER OF POINTS .LT. "KBREIT" A VALUE OF THE PHASE
C     ** SHIFT LT. "AN2" IS FOND, ALL THE PRECEDING POINTS ARE IGNORED.
C     ** KBREIT WILL BE DEFINED TIME BY TIME.
C     ** KBREIT CANNOT BE .LT. 3.
 4    I2BEG=IBEG
      NVAL=0
C     **NVAL: COUNTER FOR KBREIT.
      I3END=IEND-1
      DO 5 I3=IBEG,I3END
      IF (Y(I3+1).LE.AN2) GO TO 6
      NVAL=NVAL+1
      GO TO 5
 6    IF(NVAL.LT.KBREIT) GO TO 7
      I2END=I3+1
      GO TO 8
 7    I2BEG=I3+1
      NVAL=0
 5    CONTINUE
```

```
58        IF(NVAL.GE.KBREIT) GO TO 9
59        IF(IEND.EQ.N) GO TO 14
60        IBEG=IEND
61        GO TO 10
62   8    IEND=I2END
63   9    IBEG=I2BEG
64        CALL SLICE(X,Y,AK,AN,IBEG,IEND,YNVERS)
65   10   IEND=N
66        GO TO 4
67   14   CONTINUE
68        PRINT 11
69        WRITE(6,12)(X(K),Y(K),K=1,N,10)
70        PRINT 3
71        WRITE(6,12)(X(K),YNVERS(K),K=1,N,10)
72   11   FORMAT(1X/1X,'RESIDUAL')
73   12   FORMAT(1X,10F11.5)
74   3    FORMAT(1X/1X,'INVERTED DENSITY')
75        RETURN
76        END
```

```
 1        SUBROUTINE SLICE(X,Y,AK,AN,IBEG,IEND,YNVERS)
 2  C**** TO BE CALLED BY INVERSION SUBROUTINE FOR HILL-SHAPED PROFILES ****
 3        DIMENSION X(1000),Y(1000),YNVERS(1000)
 4        PROD=X(IBEG)*X(IEND)
 5        SUMM=X(IBEG)+X(IEND)
 6        IBEG1=IBEG+1
 7        IEND1=IEND-1
 8  C     ** TRUNCATION OF THE DO LOOP TO AVOID SQUARE ROOT  NEGATIV VALUE.
 9  C     ** FOR THE ABOVE STATEMENTS "KBREIT" CANNOT BE .LT. 3.
10        DO 1 I=IBEG1,IEND1
11        AROOTH=-X(I)**2-PROD+X(I)*SUMM
12        Y(I)=Y(I)-AN*AK*SQRT(AROOTH)
13        YNVERS(I)=YNVERS(I)+AN
14   1    CONTINUE
15        RETURN
16        END
```

APPENDIX II

```
      SUBROUTINE INVERS(X,Y,N,AK,YNVERS)
C************* INVERSION OF CRATER-SHAPED DENSITY PROFILES *************
      DIMENSION X(1000),Y(1000),YNVERS(1000)
C     **(X,Y): TABULATION OF THE PHASE SHIFT.
C     **N: NUMBER OF THE TABULATION POINTS.
C     **AK: NORMALISATION CONSTANT.
C     **YNVERS: INVERTED DENSITY.
      LOGICAL REVERT
      REVERT=.TRUE.
      Y(1)=0.
      Y(N)=0.
      YMAX=0.
      XMAX=0.
C     **YMAX: MAXIMUM OF THE PHASE SHIFT.
C     **XMAX: ABSCISSA OF THE MAXIMUM.
      DO 1 I=1,N
      YNVERS(I)=0.
      IF(YMAX.LT.Y(I)) GO TO 2
      GO TO 1
 2    YMAX=Y(I)
      XMAX=X(I)
 1    CONTINUE
C     **DETERMINATION OF THE STEP THICKNESS "AN".
C***************************
      NSTEP=500
C***************************
C     **NSTEP: "MINIMUM" NUMBER OF STEPS.
C     **AN: THICKNESS OF LOCAL DENSITY DISK.
      PROD=X(1)*X(N)
      SUMM=X(1)+X(N)
      AN=YMAX/(NSTEP*AK*SQRT(-XMAX**2-PROD+XMAX*SUMM))
      AN2=YMAX/10000.
C*********************************************************************
C     **BEGINNING OF THE "SLICING".
      IBEG=1
      IEND=N
C     **IBEG,IEND: ARE THE EXTREMES OF A SINGLE DISK.
C***************************
      J3= 3
C***************************
C     ** J3: WIDTH OF THE THE SMALLEST AREA TO BE CONSIDERED; I.E.IF
C     ** AFTER A NUMBER OF POINTS .LT. "J3" A VALUE OF THE PHASE SHIFT
C     ** LT. "AN2" IS FOND, ALL THE PRECEDING POINTS ARE IGNORED.
C     ** J3 WILL BE DEFINED TIME BY TIME.
C     ** J3 CANNOT BE .LT. 3.
 4    IBETRI=0
      IENTRI=0
C     ** IBETRI,IENTRI: OPERATIONAL PARAMETERS; THEY INDICATE THE FORM
C     ** OF THE SLICE: IF(IBETRI.EQ.0.AND.IENTRI.EQ.0) A CILINDER OF
C     ** HEIGT "AN" IS ADDED;
C     ** IF(IBETRI.EQ.1.AND.IENTRI.EQ.0) A CILINDER WITH TIANGULAR
C     ** CROSS SECTION (THE FULL HEIGT IS ON THE RIGHT HAND) IS ADDED;
C     ** IF(IBETRI.EQ.0.AND.IENTRI.EQ.1) A CILINDER WITH TIANGULAR
C     ** CROSS SECTION (THE FULL HEIGT IS ON THE LEFT HAND) IS ADDED;
C     ** IF(IBETRI.EQ.1.AND.IENTRI.EQ.1) A CILINDER WITH HEIGHT "AN"/2
C     ** IS ADDED; IN THIS CASE THE TOP OF A PARALLEL CRATER IS FOND.
C     ** THEY OPERATE IN SUBROUTINE "SLICE".
```

```
 5      IBEG1=IBEG
        IEND1=IEND
        I2BEG=IBEG
        I2END=IEND
        IA=0
        IB=0
        IA1=0
        IB1=0
        IA2=0
        IB2=0
        IA3=0
C***************************
        J50=50
C***************************
C       ** J50: MAXIMUM FORSEEN AMPLITUDE OF THE CRATER EDGE TOP.
C       ** IT WILL BE DEFINED TIME BY TIME BUT ALLWAYS .GT. J3.
C       ** IA,IB: LEFT AND RIGHT COUNTERS FOR J3.
C       ** IA1,IB1: LEFT AND RIGHT COUNTERS FOR J50.
C       ** IA2,IB2,IA3: PARAMETERS FOR LOGICAL BRANCHING.
        DO 6 I=IBEG1,IEND1
        KSTOP=INT(FLOAT(I2END-I2BEG)/2.)
C       ** KSTOP: PARAMETER TO AVOIDING THE OVERLAP OF THE TWO EXTREMES.
        IF(KSTOP.LE.J50) J50=KSTOP
        IF(KSTOP.LE.(J3+2)) J50=J3+2
        IF(IA2.EQ.1.OR.IBETRI.EQ.1) GO TO 7
        IF(Y(I).GT.AN2) GO TO 8
        IA=0
 9      IA1=IA1+1
        IF(IA1.GE.J50) GO TO 10
        IA3=1
        GO TO 7
 10     I2BEG=IBEG
        IBETRI=1
        IA=0
        IA1=0
        IA2=0
        IA3=0
        GO TO 7
 8      IA=IA+1
        IF(IA.EQ.1) I2BEG=I-1
        IF(IA.LT.J3) GO TO 9
        IBETRI =0
        IA=0
        IA1=0
        IA3=0
        IA2=1
 7      KI=IEND1-I+IBEG1
        IF(IB2.EQ.1) GO TO 11
        IF(IENTRI.EQ.1) GO TO 12
        IF(Y(KI).GT.AN2) GO TO 13
        IB=0
        IB1=IB1+1
        IF(IB1.GE.J50) GO TO 14
        IF(IA3.EQ.1) GO TO 21
 22     IF(KI.LE.(I2BEG+J3)) GO TO 16
        GO TO 6
 21     IF(KI.LE.(I+J3)) GO TO 16
```

```
115         GO TO 6
116   16    I2END=I2BEG
117         GO TO 17
118   14    I2END=IEND
119         IENTRI=1
120         IB=0
121         IB1=0
122         IB2=0
123         IF(IA3.EQ.1) GO TO 15
124         GO TO 18
125   13    IB=IB+1
126         IF(IB.EQ.1) I2END=KI+1
127         IF(IB.GT.J3) GO TO 20
128         IB1=IB1+1
129         IF(IA3.EQ.1) GO TO 21
130         GO TO 22
131   20    IB=0
132         IB1=0
133         IB2=1
134         IENTRI=0
135         IF(IA3.EQ.1) GO TO 15
136         GO TO 18
137   12    IF(IA3.EQ.1) GO TO 15
138         IF(IA2.EQ.1) GO TO 18
139         IF(I2END.LE.(I2BEG+J3)) GO TO 16
140         GO TO 19
141   11    IF(IA2.EQ.1.OR.IBETRI.EQ.1) GO TO 18
142   15    IF(I.GE.(I2END-J3)) GO TO 16
143         GO TO 6
144   18    IF(KI.LE.(I2BEG+J3)) GO TO 16
145         GO TO 19
146   6     CONTINUE
147   19    IBEG=I2BEG
148         IEND=I2END
149         IF(IBETRI.EQ.1.AND.IENTRI.EQ.1) GO TO 23
150         GO TO 24
151   23    REVERT=.NOT.REVERT
152         CALL SLICE(X,Y,AK,AN,IBEG,IEND,IBETRI,IENTRI,YNVERS)
153         IBEG=IBEG+1
154         IEND=IEND-1
155         DO 25 K=1,N
156         Y(K)=-Y(K)
157         YNVERS(K)=-YNVERS(K)
158   25    CONTINUE
159         GO TO 4
160   24    CONTINUE
161         CALL SLICE(X,Y,AK,AN,IBEG,IEND,IBETRI,IENTRI,YNVERS)
162         IF(IBETRI.EQ.0) IBEG=IBEG+1
163         IF(IENTRI.EQ.0) IEND=IEND-1
164         GO TO 5
165   17    IF(REVERT) GO TO 26
166         DO 27 K=1,N
167         Y(K)=-Y(K)
168         YNVERS(K)=-YNVERS(K)
169   27    CONTINUE
170   26    CONTINUE
171         PRINT 28
?
```

```
172   28    FORMAT(1X/1X,'RESIDUAL')
173         WRITE(6,29)(X(K),Y(K),K=1,N)
174         PRINT 3
175   3     FORMAT(1X/1X,'INVERTED DENSITY')
176         WRITE(6,29)(X(K),YNVERS(K),K=1,N)
177   29    FORMAT(1X,10F11.5)
178         RETURN
179         END
```

```
 1          SUBROUTINE SLICE(X,Y,AK,AN,IBEG,IEND,IBETRI,IENTRI,YNVERS)
 2  C***  TO BE CALLED BY INVERSION SUBROUTINE FOR CRATER-SHAPED PROFILES ***
 3          DIMENSION X(1000),Y(1000),YNVERS(1000)
 4          PROD=X(IBEG)*X(IEND)
 5          SUMM=X(IBEG)+X(IEND)
 6          IBEG1=IBEG+1
 7          IEND1=IEND-1
 8  C       ** TRUNCATION OF THE DO LOOP TO AVOID SQUARE ROOT  NEGATIV VALUE.
 9          DO 1 I=IBEG1,IEND1
10          AROOTH=-X(I)**2-PROD+X(I)*SUMM
11          IF(IBETRI.EQ.1.AND.IENTRI.EQ.0)  GO TO 2
12          IF(IBETRI.EQ.0.AND.IENTRI.EQ.1)  GO TO 3
13          IF(IBETRI.EQ.1.AND.IENTRI.EQ.1) AN=AN/2.
14          Y(I)=Y(I)-AN*AK*SQRT(AROOTH)
15          YNVERS(I)=YNVERS(I)+AN
16          IF(IBETRI.EQ.1.AND.IENTRI.EQ.1) AN=AN*2.
17          GO TO 1
18   2      TRI10=(X(I)-X(IBEG))/(X(IEND)-X(IBEG))
19          Y(I)=Y(I)-AK*AN*TRI10* SQRT(AROOTH)
20          YNVERS(I)=YNVERS(I)+AN*TRI10
21          GO TO 1
22   3      TRI01=(X(I)-X(IEND))/(X(IBEG)-X(IEND))
23          Y(I)=Y(I)-AK*AN*TRI01* SQRT(AROOTH)
24          YNVERS(I)=YNVERS(I)+AN*TRI01
25   1      CONTINUE
26          RETURN
27          END
```